
1.1 Sleuf frezen - 1226651

Programma-instellingen

Frezen sleuf Instellingen X Y Z

Veilige hoogte +150 +150 +100

Veiligheidsafstand - - +5

Voorpositie +50 +20 +100

Start-/eindpunt van de contour +50 +20 -

Bewerkingsrichting met de klok mee

Gereedschapsinstellingen

Ø T S F1 F2 DZ IZ

10 5 8900 1100 2000 -5 5

Ø) Diameter

T) Gereedschapsnummer

S) Toerental

F1) Bewerkingsaanzet

F2) Terugtrekaanzet

DZ) Max. bewerkingsdiepte/boordiepte

IZ) Verplaatsing

Oplossing

0  BEGIN PGM 1226651 MM

1  BLK FORM 0.1 Z X+0 Y+0 Z-16

2  BLK FORM 0.2  X+100  Y+100  Z+0

3  TOOL CALL 5 Z S8900 F1100

4  L  Z+100 R0 FMAX M3

5  L  X+50  Y+20  Z+5 R0 FMAX M8

6  L  Z-5 R0 F AUTO

7  L  X+30

8  CC

9  LP  PR+40  PA+108

10 CC

11 LP  PR+40  PA+36

12 CC

13 LP  PR+40  PA-36

14 CC

15 LP  PR+40  PA-108

16 L  X+50

17 L  Z+5 R0 F2000

18 L  X+150  Y+150  Z+100 R0 FMAX

19 M30

20 END PGM 1226651 MM



Contourprogrammering | Contour frezen - 1206129  1 

TNC 320 / TNC 620 / TNC 640 | HIT-leerpakket 3-assig programmeren | 6/2018 8

1.2 Contour frezen - 1206129

Programma-instellingen

Frezen buitencontour Instellingen X Y Z

Veilige hoogte +150 +150 +100

Veiligheidsafstand - - +5

Voorpositie -20 +20 +100

Start-/eindpunt van de contour +0 +25 -

Benaderings-/vrijzetlengte LEN20

Bewerkingsrichting meelopend

Gereedschapsinstellingen

Ø T S F1 F2 DZ IZ

20 10 4500 1100 2000 -5 5

Ø) Diameter

T) Gereedschapsnummer

S) Toerental

F1) Bewerkingsaanzet

F2) Terugtrekaanzet

DZ) Max. bewerkingsdiepte/boordiepte

IZ) Verplaatsing

Oplossing

0  BEGIN PGM 1206129 MM

1  BLK FORM 0.1 Z X+0 Y+0 Z-20

2  BLK FORM 0.2  X+100  Y+100  Z+0

3  TOOL CALL 10 Z S4500 F1100

4  L  Z+100 R0 FMAX M3

5  L  X-20  Y+20  Z+5 R0 FMAX M8

6  L  Z-5 R0 F AUTO

7  APPR LT  X+0  Y+25 LEN20 RL

8  L  X+25  Y+100

9  L  X+100  Y+75

10 L  X+75  Y+0

11 L  X+0  Y+25

12 DEP LT LEN20

13 L  Z+5 R0 F2000

14 L  X+150  Y+150  Z+100 R0 FMAX

15 M30

16 END PGM 1206129 MM

1.3 Contour frezen - 1214098

Programma-instellingen

Frezen contour Instellingen X Y Z

Veilige hoogte +150 +150 +100

Veiligheidsafstand - - +5

Voorpositie -20 -20 +100

Start-/eindpunt van de contour +10 +0 -

Benaderings-/vrijzetlengte LEN20

Bewerkingsrichting meelopend

Gereedschapsinstellingen

Ø T S F1 F2 DZ IZ

16 8 5600 1600 2000 -5 5

Ø) Diameter

T) Gereedschapsnummer

S) Toerental

F1) Bewerkingsaanzet

F2) Terugtrekaanzet

DZ) Max. bewerkingsdiepte/boordiepte

IZ) Verplaatsing

Oplossing

0  BEGIN PGM 1214098 MM

1  BLK FORM 0.1 Z X+0 Y+0 Z-16

2  BLK FORM 0.2  X+150  Y+100  Z+0

3  TOOL CALL 8 Z S5600 F1600

4  L  Z+100 R0 FMAX M3

5  L  X-20  Y-20  Z+5 R0 FMAX M8

6  L  Z-5 R0 F AUTO

7  APPR LT  X+10  Y+0 LEN20 RL

8  L  Y+14

9  CC  X+25  Y+25

10 C  X+10  Y+36 DR+

11 L  Y+100

12 DEP LT LEN20

13 L  Z+5 R0 F2000

14 L  X+150  Y+150  Z+100 R0 FMAX

15 M30

16 END PGM 1214098 MM

1.4 Contour frezen - 1226664

Programma-instellingen

Frezen buitencontour Instellingen X Y Z

Veilige hoogte +150 +150 +100

Veiligheidsafstand - - +5

Voorpositie +0 +70 +100

Start-/eindpunt van de contour +0 +30 -

Strategie voor benaderen en vrijzetten Cirkelbaan met tangentiële aanslui-

ting op de contour en de rechte

Radius voor benaderen en vrijzetten R10

Bewerkingsrichting meelopend

Gereedschapsinstellingen

Ø T S F1 F2 DZ IZ

20 10 4500 1700 2000 -5 5

Ø) Diameter

T) Gereedschapsnummer

S) Toerental

F1) Bewerkingsaanzet

F2) Terugtrekaanzet

DZ) Max. bewerkingsdiepte/boordiepte

IZ) Verplaatsing

Oplossing

0  BEGIN PGM 1226664 MM

1  BLK FORM 0.1 Z X-50 Y-50 Z-16

2  BLK FORM 0.2  X+50  Y+50  Z+0

3  TOOL CALL 10 Z S4500 F1700

4  L  Z+100 R0 FMAX M3

5  L  X+0  Y+70  Z+5 R0 FMAX M8

6  L  Z-5 R0 F AUTO

7  APPR LCT  X+0  Y+30 R10 RL

8  FL  Y+30  AN+0

9  FC  Y+30 DR- R42.5  CCX+0  CCY+0

10 FSELECT2

11 L  X+0

12 DEP LCT  X+0  Y+70 R10

13 L  Z+5 R0 F2000

14 L  X+150  Y+150  Z+100 R0 FMAX

15 M30

16 END PGM 1226664 MM

2.1 Kamers en sleuven frezen - 1214155

Programma-instellingen

Kamers/sleuven

(Voorbewerken + Nabewerken)

Instellingen X Y Z

Veilige hoogte +150 +150 +100

Veiligheidsafstand - - +5

2e veiligheidsafstand - - +50

Overmaat zijkant 0,2

Overmaat diepte 0,1

Baanoverlapping 0,7

Freeswijze meelopend

Gereedschapsinstellingen

Ø T S F1 F2 DZ IZ

10 5 8900 1100 2000 -5 5

Ø) Diameter

T) Gereedschapsnummer

S) Toerental

F1) Bewerkingsaanzet

F2) Terugtrekaanzet

DZ) Max. bewerkingsdiepte/boordiepte

IZ) Verplaatsing

Oplossing

0  BEGIN PGM 1214155 MM

1  BLK FORM 0.1 Z X+0 Y+0 Z-16

2  BLK FORM 0.2  X+100  Y+100  Z+0

3  TOOL CALL 5 Z S8900 F1100

4  L  Z+100 R0 FMAX M3

5  CYCL DEF 251 RECHTHOEKIGE KAMER ~

Q215=+0 ;BEWERKINGSOMVANG ~

Q218=+20 ;LENGTE 1E ZIJKANT ~

Q219=+70 ;LENGTE 2E ZIJKANT ~

Q220=+6 ;HOEKRADIUS ~

Q368=+0.2 ;OVERMAAT ZIJKANT ~

Q224=+0 ;ROTATIEPOSITIE ~

Q367=+0 ;POSITIE KAMER ~

Q207= AUTO ;AANZET FREZEN ~

Q351=+1 ;FREESWIJZE ~

Q201=-5 ;DIEPTE ~

Q202=+5 ;DIEPTEVERPLAATSING ~

Q369=+0.1 ;OVERMAAT DIEPTE ~

Q206= AUTO ;AANZET DIEPTEVERPL. ~

Q338=+0 ;VERPLAATSING NABEW. ~

Q200=+5 ;VEILIGHEIDSAFSTAND ~

Q203=+0 ;COORD. OPPERVLAK ~

Q204=+50 ;2E VEILIGHEIDSAFST. ~

Q370=+0.7 ;BAANOVERLAPPING ~

Q366=+2 ;INSTEKEN ~

Q385= AUTO ;AANZET NABEWERKEN ~

Q439=+3 ;REF. AANZET

6  L  X+30  Y+50 R0 FMAX M99

7  CYCL DEF 252 RONDKAMER ~

Q215=+0 ;BEWERKINGSOMVANG ~

Q223=+30 ;CIRKEL DIAMETER ~

Q368=+0.2 ;OVERMAAT ZIJKANT ~

Q207= AUTO ;AANZET FREZEN ~

Q351=+1 ;FREESWIJZE ~

Q201=-5 ;DIEPTE ~

Q202=+5 ;DIEPTEVERPLAATSING ~

Q369=+0.1 ;OVERMAAT DIEPTE ~

Q206= AUTO ;AANZET DIEPTEVERPL. ~

Q338=+0 ;VERPLAATSING NABEW. ~

Q200=+5 ;VEILIGHEIDSAFSTAND ~

Q203=+0 ;COORD. OPPERVLAK ~

Q204=+50 ;2E VEILIGHEIDSAFST. ~

Q370=+0.7 ;BAANOVERLAPPING ~

Q366=+1 ;INSTEKEN ~

Q385= AUTO ;AANZET NABEWERKEN ~

Q439=+3 ;REF. AANZET

8  L  X+75  Y+60 R0 FMAX M99

9  CYCL DEF 253 SLEUFFREZEN ~

Q215=+0 ;BEWERKINGSOMVANG ~

Q218=+50 ;SLEUFLENGTE ~

Q219=+12 ;SLEUFBREEDTE ~

Q368=+0.2 ;OVERMAAT ZIJKANT ~

Q374=+20 ;ROTATIEPOSITIE ~

Q367=+0 ;SLEUF POSITIE ~

Q207= AUTO ;AANZET FREZEN ~

Q351=+1 ;FREESWIJZE ~

Q201=-5 ;DIEPTE ~

Q202=+5 ;DIEPTEVERPLAATSING ~

Q369=+0.1 ;OVERMAAT DIEPTE ~

Q206= AUTO ;AANZET DIEPTEVERPL. ~

Q338=+0 ;VERPLAATSING NABEW. ~

Q200=+5 ;VEILIGHEIDSAFSTAND ~

Q203=+0 ;COORD. OPPERVLAK ~

Q204=+50 ;2E VEILIGHEIDSAFST. ~

Q366=+2 ;INSTEKEN ~

Q385= AUTO ;AANZET NABEWERKEN ~

Q439=+3 ;REF. AANZET

10 L  X+70  Y+22 R0 FMAX M99

11 L  X+150  Y+150  Z+100 R0 FMAX

12 M30

13 END PGM 1214155 MM

3.1 Boren en schroefdraad boren - 1226642

Programma-instellingen

Boren/schroefdraad Instellingen X Y Z

Veilige hoogte +150 +150 +100

Veiligheidsafstand - - +5

2e veiligheidsafstand - - +50

Gereedschapsinstellingen

Ø T S F1 F2 DZ IZ

6,8 229 6000 840 2000 -17 17

M8 264 200 - - -17 17

Ø) Diameter

T) Gereedschapsnummer

S) Toerental

F1) Bewerkingsaanzet

F2) Terugtrekaanzet

DZ) Max. bewerkingsdiepte/boordiepte

IZ) Verplaatsing

Oplossing

0  BEGIN PGM 1226642 MM

1  BLK FORM 0.1 Z X+0 Y+0 Z-16

2  BLK FORM 0.2  X+100  Y+100  Z+0

3  TOOL CALL 229 Z S6000 F840

4  L  Z+100 R0 FMAX M3

5  CYCL DEF 200 BOREN ~

Q200=+5 ;VEILIGHEIDSAFSTAND ~

Q201=-17 ;DIEPTE ~

Q206= AUTO ;AANZET DIEPTEVERPL. ~

Q202=+17 ;DIEPTEVERPLAATSING ~

Q210=+0 ;STILSTANDSTIJD BOVEN ~

Q203=+0 ;COORD. OPPERVLAK ~

Q204=+50 ;2E VEILIGHEIDSAFST. ~

Q211=+0 ;STILSTANDSTIJD ONDER ~

Q395=+1 ;REF. DIEPTE

6  CALL LBL 1

7  CALL LBL 2

8  L  Z+100 R0 FMAX

9  TOOL CALL 264 Z S200

10 L  Z+100 R0 FMAX M3

11 CYCL DEF 207 SCHR. TAPPEN GS ~

Q200=+5 ;VEILIGHEIDSAFSTAND ~

Q201=-17 ;DRAADDIEPTE ~

Q239=+1.25 ;SPOED ~

Q203=+0 ;COORD. OPPERVLAK ~

Q204=+50 ;2E VEILIGHEIDSAFST.

12 CALL LBL 1

13 L  X+150  Y+150  Z+100 R0 FMAX

14 M30

15 LBL 1

16 CYCL DEF 221 MODEL OP LIJN ~

Q225=+30 ;STARTPUNT 1E AS ~

Q226=+15 ;STARTPUNT 2E AS ~

Q237=+10 ;AFSTAND 1E AS ~

Q238=+20 ;AFSTAND 2E AS ~

Q242=+8 ;AANTAL KOLOMMEN ~

Q243=+2 ;AANTAL REGELS ~

Q224=+70 ;ROTATIEPOSITIE ~

Q200=+5 ;VEILIGHEIDSAFSTAND ~

Q203=+0 ;COORD. OPPERVLAK ~

Q204=+50 ;2E VEILIGHEIDSAFST. ~

Q301=+1 ;VERPL.VEILIGH.HOOGTE

17 LBL 0

18 LBL 2

19 CYCL DEF 220 PATROON OP CRKL ~

Q216=+70 ;MIDDEN 1E AS ~

Q217=+30 ;MIDDEN 2E AS ~

Q244=+30 ;DIAMETER STEEKCIRKEL ~

Q245=+30 ;STARTHOEK ~

Q246=+360 ;EINDHOEK ~

Q247=+60 ;HOEKSTAP ~

Q241=+6 ;AANTAL BEWERKINGEN ~

Q200=+5 ;VEILIGHEIDSAFSTAND ~

Q203=+0 ;COORD. OPPERVLAK ~

Q204=+50 ;2E VEILIGHEIDSAFST. ~

Q301=+1 ;VERPL.VEILIGH.HOOGTE ~

Q365=+0 ;TYPE VERPLAATSING

20 LBL 0

21 END PGM 1226642 MM


	TNC 320 / TNC 620 / TNC 640
	Inhoud
	1 Contourprogrammering
	1.1 Sleuf frezen - 1226651
	1.2 Contour frezen - 1206129
	1.3 Contour frezen - 1214098
	1.4 Contour frezen - 1226664

	2 Cyclusprogrammering
	2.1 Kamers en sleuven frezen - 1214155

	3 Programmeertechnieken
	3.1 Boren en schroefdraad boren - 1226642


